The TTEST Procedure

One of the basic test procedures in a first statistics course is the comparison of two population means. These population means are tested for equality using a t-test. Recall that in the two-sample t-tests, the procedures were different when the two samples were independent versus when the samples were paired or dependent. For these procedures the populations sampled were normally distributed. Other techniques are appropriate when the data are not normally distributed.

Notation:

	Population 1	Population 2
Mean	μ_1	μ_2
Variance	σ_1^2	σ_2^2

	Sample 1	Sample 2
Mean	$\bar{y}_1$	$\bar{y}_2$
Variance	s_1^2	s_2^2
Sample Size	n_1	n_2

Hypotheses	**Test Statistic**	**Reject H_0 if**	**$(1-\alpha)100\%$ CI**
H_0: $\mu_1 - \mu_2 = 0$ H_1: $\mu_1 - \mu_2 \neq 0$ Samples are **independent.** Population variances are assumed to be **equal**.	$t = \dfrac{\bar{y}_1 - \bar{y}_2}{\sqrt{s_p^2\left(\frac{1}{n_1}+\frac{1}{n_2}\right)}}$ where $s_p^2 = \dfrac{(n_1-1)s_1^2+(n_2-1)s_2^2}{n_1+n_2-2}$ is the pooled variance term.	$\lvert t\rvert \geq t_{\alpha/2,df}$ where df = $n_1 + n_2 - 2$	$(\bar{y}_1 - \bar{y}_2) \pm t_{\alpha/2,df}\sqrt{s_p^2\left(\frac{1}{n_1}+\frac{1}{n_2}\right)}$
H_0: $\mu_1 - \mu_2 = 0$ H_1: $\mu_1 - \mu_2 \neq 0$ Samples are **independent.** Population variances are assumed to be **not equal.**	$t = \dfrac{\bar{y}_1 - \bar{y}_2}{\sqrt{\left(\frac{s_1^2}{n_1}+\frac{s_2^2}{n_2}\right)}}$	$\lvert t\rvert \geq t_{\alpha/2,df*}$ where $df* = \dfrac{\left(\frac{s_1^2}{n_1}+\frac{s_2^2}{n2}\right)^2}{\left[\frac{\left(s_1^2/n_1\right)^2}{(n_1-1)}+\frac{\left(s_2^2/n2\right)^2}{(n_2-1)}\right]}$	$(\bar{y}_1 - \bar{y}_2) \pm t_{\alpha/2,df*}\sqrt{\left(\frac{s_1^2}{n_1}+\frac{s_2^2}{n_2}\right)}$
H_0: $\mu_1 - \mu_2 = 0$ H_1: $\mu_1 - \mu_2 \neq 0$ Samples are **dependent.** For each pair of observations x and y, compute d = x – y.	$t = \dfrac{\bar{d}}{\sqrt{s_d/n}}$ where $\bar{d}$ is the average difference, s_d is the standard deviation of the differences, and n is the number of pairs of observations.	$\lvert t\rvert \geq t_{\alpha/2,df}$ where df = n – 1	$\bar{d} \pm t_{\alpha/2,df}\sqrt{s_d/n}$

For the independent samples t-test, there are two procedures: one when the variances are equal and another when the variances are not equal. There is a test for the equality of two population variances to determine which method of comparing the population means is appropriate.

Hypotheses	**Test Statistic**	**Reject H_0 if**
H_0: $\sigma_1^2 = \sigma_2^2$ H_1: $\sigma_1^2 \neq \sigma_2^2$	$F = \frac{s_1^2}{s_2^2}$ Although this is a two-sided test, the test statistic is usually calculated with the larger sample variance in the numerator and only the right side of the rejection region is then specified.	$F \geq F_{\alpha/2, n_1-1, n_2-1}$

The TTEST procedure tests the hypothesis that the means of two independent and normally distributed groups are equal. The test statistics under the assumptions of both equal and unequal variance are computed. This procedure also tests the equality of the two group variances.

The syntax of the TTEST procedure is:

```
PROC TTEST  DATA=setname  <options> ;
CLASS variable ;
PAIRED pairlists ;
VAR variable list  ;
RUN;
```

No statement can be used more than once. The statements can appear in any order after the PROC TTEST statement. BY and WHERE statements are optional.

Options on the TTEST procedure include:

ALPHA=p — $0 < p < 1$, this value is used to determine the level of confidence used in the confidence interval calculations.

CI=EQUAL — an equal tailed confidence interval for σ is computed.

CI=UMPU — an interval based on the uniformly most powerful test of σ

CI=NONE — no confidence interval for σ is printed. CI's for the means and the difference between them will still be computed.

The default setting for all confidence intervals is 95% unless otherwise specified using an ALPHA option. CI=EQUAL and CI=UMPU can both be specified. For σ, two confidence interval estimates will be computed and printed in the TTEST output.

PLOTS = (*list*) — ODS Graphics plots requested. Available graphics are: ALL, NONE, HISTOGRAM, BOXPLOT, INTERVAL, QQ, PROFILES, AGREEMENT, SUMMARY. When requesting more than one ODS Graphic, enclose the list in parenthesis, such as PLOTS=(QQ PROFILES)

For paired data the defaults plots are:

```
PLOTS=(HISTOGRAM BOXPLOT INTERVAL QQ PROFILES AGREEMENT)
```

SIDES = 2 | L | U (or SIDED or SIDE)

SIDES=2 specifies the two-sided test and confidence interval for the mean.

SIDES = L specifies the *lower* one-sided tests and the confidence interval from negative infinity to the upper confidence bound for the mean.

SIDES = U specifies the *upper* one-sided tests and the confidence interval from the lower confidence bound for the mean and positive infinity.

CLASS Statement

A CLASS statement giving the name of the grouping variable must accompany the PROC TTEST statement when conducting an independent sample t-test. The grouping variable must have two, and only two, levels, either numeric or character.

BY Statement

A BY statement is optional can be used with PROC TTEST to obtain separate analyses on observations in groups defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables.

PAIRED Statement

The variables in the PAIRED statement are to be compared using a paired t-test. Differences are computed using the variable on the left minus the variable on the right in the PAIRED statement. The CLASS statement and the VAR statement **cannot** be used with the PAIRED statement. The syntax of this statement is summarized as follows:

PAIRED statements	Yield these comparisons
PAIRED A*B ;	A – B
PAIRED A*B C*D ;	A – B and C – D
PAIRED (A B) * (C D);	A – B, A – C, B – C, B – D
PAIRED (A B) * (B C);	A – B, A – C, B – C

VAR Statement

The VAR statement names the variables whose means are to be compared using an independent t-test. If the VAR statement is omitted, all numeric variables in the input data set (except a numeric variable in the CLASS statement) are included in the analysis.

Objective 3: Paired Data Example Subjects are measured for improvement after receiving a treatment. Run the program below to analyze the BEFORE vs AFTER treatment data. Compute a 99% confidence interval for the true difference between the means.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
DATA one;
INPUT subject before after ;
DATALINES;
1 138 324
2 284 520
3 234 318
4 132 220
5 183 232
```

```
PROC TTEST DATA=one CI=NONE ALPHA=0.01;
PAIRED before*after ;
RUN;
QUIT;
```

1

The TTEST Procedure

Difference: before - after

N	Mean	Std Dev	Std Err	Minimum	Maximum
5	-128.6	78.7452	35.2159	-236.0	-49.0000
n	$\bar{d}$	s_d	s_d / sqrt(n)		

Mean	99% CL Mean	
-128.6	-290.7	33.5374

The observed average difference is -128.6 and the 99% CI for the mean difference is (-290.7, 33.537).

DF	t Value	Pr > |t|
4	-3.65	0.0217

There is not a significant difference between the before and after measurement means (α = 0.01, t_4 = -3.65, p = 0.0217).

If ODS Graphics are enabled, then all of the Figures 1 – 4 result.

Figure 1: Summary panel – the HISTOGRAM and BOXPLOTS together in a single panel.



ODS Graphics Names

`HISTOGRAM`

With overlaid normal density

`BOXPLOT`

`INTERVAL`

overlaid interval band (shaded)

Figure 2: The first observation is connected to the second observation for each subject or pair. Note the means for the pairs is also plotting on the bold, red dashed line.



ODS Graphics Names

`PROFILES`

for paired data only.

Figure 3: The second response (see the PAIRED statement) is plotted on the vertical axis and the first response is plotted on the horizontal axis. The farther from the 45° reference line, the more likely a significant difference is detected.



ODS Graphics Names

AGREEMENT

for paired data only.

Figure 4: This is a normal Q-Q plot. Data that are normally distributed lie "close" to the 45° reference line.



ODS Graphics Names

QQ

for the paired differences.

In this program, one could also have written the PAIRED statement as:
PAIRED after – before ;

and the resulting output would be:

The TTEST Procedure

Difference: after - before

N	Mean	Std Dev	Std Err	Minimum	Maximum
5	128.6	78.7452	35.2159	49.0000	236.0

Mean	99% CL Mean	
128.6	-33.5374	290.7

DF	t Value	Pr > |t|
4	3.65	0.0217

Note the values that are the opposite sign of the first scenario. The conclusion for the hypothesis is the same.

Compare the ODS Graphics obtained with Figures 1 through 4 above.

Objective 4: Rerun the program with ODS Graphics enabled but with a PLOTS option on the PROC TTEST statement. Specifically,

i.
```
PROC TTEST DATA=one ALPHA=0.01 PLOTS(ONLY)=(HISTOGRAM BOXPLOT);
PAIRED before*after ;
```

ii.
```
PROC TTEST DATA=one CI=NONE ALPHA=0.01 PLOTS(ONLY)=(BOXPLOT
INTERVAL)  SIDED=L;
PAIRED before*after ;
```

PLOTS(ONLY) will restrict the plots to those requested. When ODS Graphics are enabled, all possible plots are printed. If you omit ONLY, you will still receive all four of the plots illustrated above.

Objective 5: Independent t-test Example Two instructional programs are to be compared. Students independently participate in one of the two programs. Scores at the completion of the instructional period are recorded for each of the students. Information is given in the table below.

Program A	Program B
71 82 88 64 59 78	65 88 92 76 87 89
72 81 83 66 83 91	85 90 81 91 78 81
79 70	86 82 73 79

Assume that the two populations of computing time are normally distributed. Is there evidence that one program yields higher results than the other? Run the following program to analyze the independent samples. Compute a 95% confidence interval for the mean. Identify whether the equal variance or unequal variances approach to the analysis is more appropriate.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';

DATA one;
INPUT program $ score @@;
DATALINES;
A 71 A 82 A 88 A 64 A 59 A 78 A 72
A 81 A 83 A 66 A 83 A 91 A 79 A 70
B 65 B 88 B 92 B 76 B 87 B 89 B 85
B 90 B 81 B 91 B 78 B 81 B 86 B 82
B 73 B 79
;
PROC TTEST DATA=one CI=EQUAL;
CLASS program;
VAR score;
TITLE 't-test for the Difference Between Two Independent Means';

RUN;

QUIT;
```

t-test for the Difference Between Two Independent Means

The TTEST Procedure

Variable: score

program	N	Mean	Std Dev	Std Err	Minimum	Maximum
A	14	76.2143	9.4069	2.5141	59.0000	91.0000
B	16	82.6875	7.3277	1.8319	65.0000	92.0000
Diff (1-2)		-6.4732	8.3576	3.0586		

program	Method	Mean	95% CL Mean		Std Dev	95% CL Std Dev	
A		76.2143	70.7829	81.6456	9.4069	6.8195	15.1549
B		82.6875	78.7828	86.5922	7.3277	5.4130	11.3411
Diff (1-2)	**Pooled**	-6.4732	-12.7384	-0.2080	8.3576	6.6324	11.3033
Diff (1-2)	**Satterthwaite**	-6.4732	-12.8867	-0.0597			

Method	Variances	DF	t Value	Pr > |t|
Pooled	Equal	28	-2.12	0.0433
Satterthwaite	Unequal	24.487	-2.08	0.0481

Equality of Variances				
Method	**Num DF**	**Den DF**	**F Value**	**Pr > F**
Folded F	13	15	1.65	0.3527


Distribution of score
A
B
Percent
Percent
program
score
Normal
Kernel


Q-Q Plots of score
A
B
score
Quantile

Mean for program A is 76.214; mean for program B is 82.688. The difference between means (A – B) is -6.473. The std deviations are 9.4069 for A and 7.3277 for B. Std error is std dev / sqrt(n) .

The t-tests are for H_0: $\mu_A = \mu_B$ vs H_0: $\mu_A \neq \mu_B$. t_{28} = -2.12 is calculated under the equal variance assumption, and $t_{24.5}$ = -2.08 is calculated under the unequal variance assumption. SAS does not determine which it appropriate. You must do that.

Equality of Variances: This is the F-test for testing H_0: $\sigma^2_1 = \sigma^2_2$ vs H_1: $\sigma^2_1 \neq \sigma^2_2$. Here $F_{13,15}$ = 1.65.

Since a 95% CI was specified, then we'll use α = 0.05 for all tests.

The population variances for the scores are not significantly different (α = 0.05, $F_{13,15}$= 1.65, p=0.3527). The appropriate t-test to compare the population means assumes that the variances are equal. There is a significant difference between the two population means (α = 0.05, $t_{-2.12}$ = 0.0433). Since this difference is significant, we can further conclude that Program B has the larger population mean score.

The ODS Graphics for that are the default for TTEST for independent samples are HISTOGRAM, BOXPLOT, and QQ.

Objective 6: Independent t-test Example Rerun Objective 5 with the SIDED = L option and observe the output. Repeat using the SIDED = U option. How are the differences computed? Does this make a difference in the results of the t-test? How do you control this?