

**Output Delivery System (ODS)** – ODS enhances your ability to manage output from various SAS procedures. The appearance of the output can be controlled, SAS data set can be created and managed, and RTF, PDF, and HTML files can be created from your results.

### ODS Table Names

Whenever a procedure that produces output is run in SAS the result is a series of tables of information. Each of these tables can be printed, suppressed, output to a SAS data set or written in HTML or other format. SAS refers to each of these tables of results by a SAS-assigned ODS table name. There is a different set of ODS table names for every procedure producing SAS output. So it is not feasible to memorize or know them all. Two ways to determine the names of these tables are by

1. Tracing the SAS results to obtain the table name
2. Consulting SAS Help and Documentation (within SAS) for the procedure and see the list of ODS table names SAS lists.

#### 1. *Tracing SAS Output*

When a program is run, the results can be traced for ODS table names by including the following SAS statement:

ODS TRACE ON ;

This command can be inserted into a SAS program or run as a single line SAS program. Once the trace is "turned on" all subsequent SAS procedures that run will be traced. That is, when you run a program, in addition to the usual feedback in the SAS log you will also see a list of ODS table names *available for that current run of the procedure*. The trace can appear in the output listing if the command is ODS TRACE ON / LISTING ;

ODS TRACE ON ; is a global command. That is, once the trace is active, all subsequent SAS programs and procedures submitted during that SAS session will be traced until the statement

ODS TRACE OFF;

is submitted or SAS is closed. ODS TRACE OFF; can be submitted in a SAS program or run as a single line program.

A recommendation is to place ODS TRACE ON; at the top of a program to be traced and place ODS TRACE OFF; after the final RUN statement and before the QUIT statement at the end of the SAS program. In this way, the trace is not left on creating lengthier log or output listings. (No RUN statement is needed after ODS TRACE OFF;, but no error will result if a RUN statement is placed before the QUIT statement.)

#### 2. *SAS Help and Documentation*

A list of ODS table names can be found in SAS Help and Documentation for most procedures. After accessing SAS Help and Documentation, select the **Index** tab and locate the procedure for which you need the ODS table names. Within the procedure's available topics one would look for ODS table names and select the item. Here are listed the *all the available ODS table names* for that procedure (more information than in a trace) and the options needed in the programming to produce those tables. (ODS table names can also be

found using other approaches in SAS Help and Documentation, but the method here seems more general and therefore applicable to more procedures.)

**Objective 1:** Find out the ODS table names for the UNIVARIATE procedure in the following analysis by using a trace. Trace the procedure in the log window.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE ON;

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one PLOTS;
VAR a ;
RUN;

ODS TRACE OFF;
QUIT;
```

Here the UNIVARIATE procedure is run. The ODS statements first TRACE the available SAS data sets. This trace is printed in the log window. When run, this program SAS identifies the data tables Moments, BasicMeasures, TestsForLocation, Quantiles, ExtremeObs, and Plots. One of the tables in the SAS output is "Moments" and the corresponding trace information is:

SAS log	SAS output listing
Output Added: ----- Name: Moments Label: Moments Template: base.univariate.Moments Path: Univariate.a.Moments	Moments  N 3 Sum Weights 3 Mean 54.3333333 Sum Observations 163 Std Deviation 28.9194283 Variance 836.333333 Skewness 1.40262706 Kurtosis . Uncorrected SS 10529 Corrected SS 1672.66667 Coeff Variation 53.2259417 Std Error Mean 16.6966397
The ODS table name is identified in the first line of the trace: <b>Name: Moments</b>	Notice the heading for the table in this UNIVARIATE output is Moments. Typically, the ODS table name will very closely match the table heading in the output. When a table has a multi-word heading, the ODS data table name will run the words of the table name together and may truncate some of the words as shown in paragraph preceding this table.

**Objective 2:** Trace the program in Objective 1 in the output listing.

Change ODS TRACE ON; at the top of the program to ODS TRACE ON / LISTING ;

Observe that the trace information occurs immediately before the corresponding table in the output listing.

Output Added:

-----

Name: Moments  
 Label: Moments  
 Template: base.univariate.Moments  
 Path: Univariate.a.Moments  
 -----

Moments			
N	3	Sum Weights	3
Mean	54.3333333	Sum Observations	163
Std Deviation	28.9194283	Variance	836.333333
Skewness	1.40262706	Kurtosis	.
Uncorrected SS	10529	Corrected SS	1672.66667
Coeff Variation	53.2259417	Std Error Mean	16.6966397

---

**Objective 3:** Find the list of all the possible ODS table names produced for the UNIVARIATE Procedure using SAS Help and Documentation.

The screenshot shows the SAS Help and Documentation window. The left pane displays a search results list with 'UNIVARIATE procedure, ODS table names' selected. The right pane shows the 'The UNIVARIATE Procedure' page, specifically the 'ODS Table Names' section. Below this section is a table listing the ODS tables produced by the PROC UNIVARIATE statement.

**ODS Table Names**

PROC UNIVARIATE assigns a name to each table that it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets.

**Table 3.69: ODS Tables Produced with the PROC UNIVARIATE Statement**

ODS Table Name	Description	Option
BasicIntervals	Confidence intervals for mean, standard deviation, variance	CIBASIC
BasicMeasures	Measures of location and variability	Default
ExtremeObs	Extreme observations	Default
ExtremeValues	Extreme values	NEXTRVAL=
Frequencies	Frequencies	FREQ
LocationCounts	Counts used for sign test and signed rank test	LOCCOUNT
MissingValues	Missing values	Default, if missing values exist
Modes	Modes	MODES
Moments	Sample moments	Default
Plots	Line printer plots	PLOTS
Quantiles	Quantiles	Default
RobustScale	Robust measures of scale	ROBUSTSCALE
SSPlots	Line printer side-by-side box plots	PLOTS (with BY statement)
TestsForLocation	Tests for location	Default
TestsForNormality	Tests for normality	NORMALTEST

In the above screen capture, only part of the list of available ODS table names for PROC UNIVARIATE appears.

## Controlling the Appearance of the Output

When submitting a SAS procedure, some or all of the tables in the results may not be needed. Perhaps a procedure is run so that an output SAS data set is created. Not all procedures have a NOPRINT option such as PROC MEANS and PROC UNIVARIATE, but it would produce less unnecessary output if the output results were suppressed wholly or in part.

### 1. *Suppressing/Expressing All Information in the HTML Output for a Procedure/Program*

When a program is submitted, the output can be suppressed for all subsequent programs and procedures by using the following SAS statement:

```
ODS HTML CLOSE;           (or ODS HTML EXCLUDE ALL;
                           or ODS HTML SELECT NONE; )
```

This command can be inserted into a SAS program or run as a single line SAS program.

ODS HTML CLOSE; is a global command. That is, once the HTML is off, all subsequent SAS programs and procedures submitted during that SAS session will have suppressed HTML output until the statement

```
ODS HTML;                 (or ODS HTML SELECT ALL;
                           or ODS HTML EXCLUDE NONE; )
```

is submitted or SAS is closed. ODS HTML; can be submitted as part of a SAS program or run as a single line program. ODS HTML; is a global command.

If one chooses to print results to the LISTING Output rather than the Results Viewer, the LISTING Output can be enabled by the global statement ODS LISTING; To disable the LISTING Output, the global statement is ODS LISTING CLOSE;

Recall that the LISTING Output does not contain any high resolution graphics nor ODS Graphics, but a text only form of the output. A recommendation is to place ODS LISTING CLOSE; in a program prior to procedures that are to have suppressed LISTING Output and place ODS LISTING; after a RUN statement and prior to any procedures which are not to have suppressed LISTING Output or prior to the QUIT statement. (No RUN statement is needed after ODS LISTING, but no error will result if a RUN statement is included.)

Ensuring that the LISTING Output is not suppressed when a SAS programming session begins, the ODS LISTING; statement could be included at the top of a program as regularly as some use Display Manager (DM) commands, OPTIONS, and TITLE and FOOTNOTE resets. For example:

```
DM 'LOG; CLEAR; OUTPUT; CLEAR; ODSRESULTS; CLEAR; ' ;
TITLE; FOOTNOTE;
ODS LISTING;
```

may be at the top of programs. OUTPUT; CLEAR; in the DM command, clears the LISTING Output, while ODSRESULTS; CLEAR; in the DM command clears the Results Viewer (HTML Output). Some SAS programmers have adopted a set of routine commands to be included at the beginning of a SAS program so that they know all their preferred settings are in place with every SAS program they submit. The example is just a sample of what a SAS programmer could have in place.

## 2. *Suppressing Parts of HTML Output for a Procedure*

Some procedures generate multiple pages or results. It may not be desirable or necessary to print all of the possible output for a procedure. Within a block of statements, one can select one or more tables for inclusion in the results or exclude one or more tables from the results.

The general syntax for the selecting tables to be printed is:

```
ODS HTML SELECT ODStablenameslist ;
```

The general syntax for excluding tables to be printed is:

```
ODS HTML EXCLUDE ODStablenamelist ;
```

where the ODS table names listed need only be separated by a space (in both of these statements).

One or more tables can be specified. Either SELECT a few tables or EXCLUDE a few tables, whichever is a more efficient method. The chosen statement would occur within the block of statements associated with a SAS procedure. If the ODS table names are not known, the program would either need to be first traced or the SAS Help and Documentation would have to be consulted to obtain that information.

**Objective 4:** Modify the program in Objective 1 by replacing the PLOTS options with the NORMAL option. Print only the Tests for Location and the Tests for Normality in the HTML output for all 3 variables. Use the trace information to identify the ODS table names needed.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE ON;

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one NORMAL;
VAR a b c;
ODS HTML SELECT TESTSFORLOCATION TESTSFORNORMALITY ;
RUN;

ODS TRACE OFF;
QUIT;
```

Notice that the UNIVARIATE HTML Output is much shorter than usual, containing only those two items requested for each variable in the VAR statement.

**Objective 5:** Modify the program in Objective 4 so that the table of Extreme Observations is excluded from the HTML Output.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE ON;
```

```

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one NORMAL;
VAR a ;
ODS LISTING EXCLUDE EXTREMEOBS ; ;
RUN;

ODS TRACE OFF;
QUIT;

```

### Creating SAS Data Sets from the Output

Each of the tables in the Output is referred to by an ODS table name, and these tables can also be recovered as SAS data sets, either temporary or permanent.

The general syntax for this is

ODS OUTPUT *ODStablename*=*SAS-data-set*;

where *SAS-data-set* can be either a one- or two-level SAS-data-set name.

The ODS OUTPUT statement is placed within a procedure in order to recover the procedure's output results in a SAS data set. Multiple ODS table names and SAS data sets can be defined in a single ODS OUTPUT statement.

**Objective 6:** Using the program in Objective 4, include the statement that would save the Moments information into a temporary SAS data set called *basicstats* and save the Tests for Locations as a temporary SAS data set called *meantests*. Print both recovered SAS data sets.

```

DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE ON;

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one NORMAL;
VAR a b c;
ODS HTML SELECT TESTSFORLOCATION TESTSFORNORMALITY ;
ODS OUTPUT MOMENTS=basicstats TESTSFORLOCATION=meantests;

PROC PRINT DATA=basicstats;
PROC PRINT DATA=meantests;
RUN;

ODS TRACE OFF;

```

QUIT;

It is worth noting that only two tables were printed as a result of the UNIVARIATE procedure and ODS HTML SELECT. The Moments table was not printed in the UNIVARIATE procedure but that information can still be recovered as a SAS data set. This implies that the full procedure "runs" but only what is selected in ODS HTML is printed.

Observe the SAS data set *basicstats* as it is printed in the output listing:

Obs	VarName	Label1	cValue1	nValue1	Label2	cValue2	nValue2
1	a	N	3	3.000000	Sum Weights	3	3.000000
2	a	Mean	54.3333333	54.333333	Sum Observations	163	163.000000
3	a	Std Deviation	28.9194283	28.919428	Variance	836.333333	836.333333
4	a	Skewness	1.40262706	1.402627	Kurtosis	.	.
5	a	Uncorrected SS	10529	10529	Corrected SS	1672.66667	1672.666667
6	a	Coeff Variation	53.2259417	53.225942	Std Error Mean	16.6966397	16.696640
7	b	N	3	3.000000	Sum Weights	3	3.000000
8	b	Mean	69	69.000000	Sum Observations	207	207.000000
9	b	Std Deviation	18.5202592	18.520259	Variance	343	343.000000
10	b	Skewness	-1.457863	-1.457863	Kurtosis	.	.
11	b	Uncorrected SS	14969	14969	Corrected SS	686	686.000000
12	b	Coeff Variation	26.8409553	26.840955	Std Error Mean	10.6926766	10.692677
13	c	N	3	3.000000	Sum Weights	3	3.000000
14	c	Mean	73.3333333	73.333333	Sum Observations	220	220.000000
15	c	Std Deviation	30.664855	30.664855	Variance	940.333333	940.333333
16	c	Skewness	-1.2775942	-1.277594	Kurtosis	.	.
17	c	Uncorrected SS	18014	18014	Corrected SS	1880.66667	1880.666667
18	c	Coeff Variation	41.8157114	41.815711	Std Error Mean	17.7043623	17.704362

and as it appears using View Table.

	VarName	Label1	cValue1	nValue1	Label2	cValue2	nValue2
1	a	N	3	3.000000	Sum Weights	3	3.000000
2	a	Mean	54.3333333	54.333333	Sum Observations	163	163.000000
3	a	Std Deviation	28.9194283	28.919428	Variance	836.333333	836.333333
4	a	Skewness	1.40262706	1.402627	Kurtosis	.	.
5	a	Uncorrected SS	10529	10529	Corrected SS	1672.66667	1672.666667
6	a	Coeff Variation	53.2259417	53.225942	Std Error Mean	16.6966397	16.696640
7	b	N	3	3.000000	Sum Weights	3	3.000000
8	b	Mean	69	69.000000	Sum Observations	207	207.000000
9	b	Std Deviation	18.5202592	18.520259	Variance	343	343.000000
10	b	Skewness	-1.457863	-1.457863	Kurtosis	.	.
11	b	Uncorrected SS	14969	14969	Corrected SS	686	686.000000
12	b	Coeff Variation	26.8409553	26.840955	Std Error Mean	10.6926766	10.692677
13	c	N	3	3.000000	Sum Weights	3	3.000000
14	c	Mean	73.3333333	73.333333	Sum Observations	220	220.000000
15	c	Std Deviation	30.664855	30.664855	Variance	940.333333	940.333333
16	c	Skewness	-1.2775942	-1.277594	Kurtosis	.	.
17	c	Uncorrected SS	18014	18014	Corrected SS	1880.66667	1880.666667
18	c	Coeff Variation	41.8157114	41.815711	Std Error Mean	17.7043623	17.704362



Examine the values in the *meantests* in the same ways.

**Objective 7:** Using ODS statements only, suppress all of the HTML Output information for the program in Objective 4 and recover the Basic 99% Confidence Intervals and the Tests for Normality in SAS data sets called *ci99* and *normaltests*, respectively. Note you need the CIBASIC, ALPHA=0.01, and NORMAL options on the PROC UNIVARIATE statement.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE ON;

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=ONE CIBASIC ALPHA=0.01 NORMAL;
VAR a b c ;
ODS OUTPUT BASICINTERVALS=ci99 TESTSFORNORMALITY=normaltests;
ODS HTML SELECT NONE ;
RUN;


ODS TRACE OFF;
QUIT;
```

No HTML Output is created. But examine the SAS data sets in View Table.

	VarName	Goodness-of-Fit Test	Label for Goodness-of-Fit Statistic	Value of Goodness-of-Fit Statistic	p-value Label	Sign of p-value	p-value
1	a	Shapiro-Wilk	W	0.904245	Pr < W		0.3991
2	a	Kolmogorov-Smimov	D	0.306239	Pr > D	>	0.1500
3	a	Cramer-von Mises	W-Sq	0.051495	Pr > W-Sq	>	0.2500
4	a	Anderson-Darling	A-Sq	0.298836	Pr > A-Sq	>	0.2500
5	b	Shapiro-Wilk	W	0.892857	Pr < W		0.3631
6	b	Kolmogorov-Smimov	D	0.313938	Pr > D	>	0.1500
7	b	Cramer-von Mises	W-Sq	0.054314	Pr > W-Sq	>	0.2500
8	b	Anderson-Darling	A-Sq	0.312211	Pr > A-Sq	>	0.2500
9	c	Shapiro-Wilk	W	0.92547	Pr < W		0.4719
10	c	Kolmogorov-Smimov	D	0.290375	Pr > D	>	0.1500
11	c	Cramer-von Mises	W-Sq	0.046252	Pr > W-Sq	>	0.2500
12	c	Anderson-Darling	A-Sq	0.274126	Pr > A-Sq	>	0.2500

	VarName	Parameter	Estimate	Lower 99% Confidence Limit	Upper 99% Confidence Limit
1	a	Mean	54.33333	-111.37820	220.04486
2	a	Std Deviation	28.91943	12.56379	408.47050
3	a	Variance	836.33333	157.84886	166848
4	b	Mean	69.00000	-37.12314	175.12314
5	b	Std Deviation	18.52026	8.04596	261.58814
6	b	Variance	343.00000	64.73753	68428
7	c	Mean	73.33333	-102.37969	249.04635
8	c	Std Deviation	30.66486	13.32208	433.12366
9	c	Variance	940.33333	177.47773	187596

In the UNIVARIATE procedure the variable names (column names) in these SAS data sets are 'SAS selected' and do not contain spaces or special characters. For some ODS tables column headings may have a multiword variable (column) name appearing in View table. To find out the variable names either PRINT the SAS data SET without a LABELS option or change the view in View Table to show column names rather than column labels; that is, select **View – Column Names** while in View Table as shown next.



	VarName	Parameter	Estimate	LowerCL	UpperCL
1	a	Mean	54.33333	-111.37820	220.04486
2	a	Std Deviation	28.91943	12.56379	408.47050
3	a	Variance	836.33333	157.84886	166848

## Some Suggestions

ODS TRACE ON; enables the trace of a program so that one can see the ODS tables names that can be selected, excluded, or saved as a SAS data set. Once one has identified the names of the ODS tables, it is recommended that the trace be turned off. As the program grows, the length of the log will grow due to trace of all of the procedures in a program. Additionally, if the program has one or more procedures with a BY statement, the trace gets very repetitious as the all trace identifications appear for each level of the BY variable(s) specified.

## Create HTML files from the Output

Notice in the program above the ODS OUTPUT statement does not influence what tables are included in the HTML output. ODS OUTPUT specifies the parts of the output listing to be saved as data tables.

A file destination can be specified by modifying the ODS HTML command to include a filename or file reference. That is,

```
ODS HTML FILE='full file path' ;
```

or

```
FILENAME fileref 'full file path' ;
```

```
ODS HTML FILE=fileref;
```

Can be inserted before the procedures that one wishes to recover the HTML Output as a file. After the final RUN statement, one closes the HTML communication with the statement:  
ODS HTML CLOSE;

**Objective 8:** Create an HTML file, check8.html, containing the complete output from the UNIVARIATE procedure in Objective 7. In the FILENAME statement, change the directory or folder information to correspond a directory you are using or have available.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';
ODS TRACE OFF; *once the ODS Table Names are known, turn the trace off.
```

```
*specify a directory and filename for the file to be created. ;
FILENAME obj8 'f:\check8.html';
ODS HTML FILE=obj8 ;
```

```
DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one NORMAL;
VAR a b c;
ODS HTML SELECT TESTSFORLOCATION TESTSFORNORMALITY ;
ODS OUTPUT MOMENTS=basicstats TESTSFORLOCATION=meantests;

RUN;

ODS HTML CLOSE;
QUIT;
```

In addition to being able create HTML, PDF files can also be created and later viewed using Adobe Acrobat Reader or similar software. The syntax is like that for the previous file creation:

```
ODS PDF FILE=fileref or "file path and filename" ;
to open the communication and
```

```
ODS PDF CLOSE;
to close the communication.
```

**Objective 9:** Create a PDF file called check9.pdf from the complete UNIVARIATE output in Objective 7.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';

ODS PDF FILE='f:\check10.pdf';

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one CIBASIC ALPHA=0.01 NORMAL;
VAR a b c;
RUN;

ODS PDF CLOSE;
QUIT;
```

View the resulting PDF file. If the statement: ODS PDF CLOSE; appeared before the RUN statement, no pdf file would be produced. This is because the pdf connection closed before the RUN request for the UNIVARIATE procedure.

In addition to HTML and PDF format files, RTF and other PostScript files can be created. ODS statements to open the communication and close the communication are needed just as above. Such as, ODS RTF FILE=*fileref* and ODS RTF CLOSE.

These steps are just the beginning of the output files that can be created using ODS. Advanced programming with ODS allows you to format the contents of the newly created files, such as column alignment, headers and footers and much more.

The SELECT and EXCLUDE actions are also available for the ODS HTML, ODS PDF, and other file creation statements. This allows one to recover part(s) of the results in an external file.

**Objective 10:** Modify the program in Objective 9 by adding statements that will create a PDF file check10.pdf.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';

ODS PDF FILE='f:\check10.pdf';

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one CIBASIC ALPHA=0.01 NORMAL;
```

```

VAR a b c;
RUN;

ODS HTML SELECT TESTSFORLOCATION TESTSFORNORMALITY ;
ODS OUTPUT MOMENTS=basicstats TESTSFORLOCATION=meantests;

RUN;

ODS PDF CLOSE;
QUIT;

```

Note the pdf file contains all of the UNIVARIATE output despite the ODS HTML SELECT action. This is because the SELECT action is applicable only to the HTML Output. To select the same ODS table names for inclusion in an external file, the statement ODS PDF SELECT testsforlocation testsfornormality ;

would need to appear within the UNIVARIATE procedure.

**Objective 11:** Modify the program in Objective 10 to select only the Tests for Location and Tests for Normality for inclusion in the pdf file.

```

DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ';

ODS PDF FILE='f:\check10.pdf';

DATA one;
INPUT a b c ;
DATALINES;
44 48 39
87 83 98
32 76 83
;
PROC UNIVARIATE DATA=one CIBASIC ALPHA=0.01 NORMAL;
VAR a b c;
RUN;

ODS HTML SELECT TESTSFORLOCATION TESTSFORNORMALITY ;
ODS OUTPUT MOMENTS=basicstats TESTSFORLOCATION=meantests;
ODS PDF SELECT TESTSFORLOCATION TESTSFORNORMALITY ;

PROC PRINT DATA=basicstats;
PROC PRINT DATA=meantests;
RUN;

ODS PDF CLOSE;
QUIT;

```

The ODS HTML and ODS OUTPUT statements do not affect the pdf file creation.

Notice also when you run the program in Objective 11, the results of the two PRINT procedures are also in the pdf file. ODS PDF SELECT applies to the current procedure only. How could the results of the two PRINT procedures be included in the HTML Output but not the pdf file?

*Answer:* Move the ODS PDF CLOSE statement prior to the two print procedures but following the ODS PDF SELECT statement.

With ODS one can control output to multiple "targets," such as, PDF, LISTING, data tables, and more.

Note that many of the controls require you to 'open' and 'close' the ODS action (TRACE, HTML, PDF, RTF, . . .) and other ODS statements are applicable to the current procedure (SELECT, EXCLUDE, OUTPUT, . . .) . Many of the ODS statements have additional actions not covered here, such as page formatting.

Many SAS procedures now have some high-resolution graphics available. ODS also has a GRAPHICS control which has been investigated in some of the procedures covered earlier.

This has been a very basic introduction to ODS. Hopefully you'll discover the utility of ODS in your applications.