

The PLOT Procedure

The PLOT procedure plots one variable versus another using horizontal and vertical axes. The plot is a low-resolution plot. You can use character or numeric values as plotting symbols. You can specify the dimension of the plot or use reference lines. The syntax of the PLOT procedure is:

```
PROC PLOT DATA=setname <options> ;
BY variablelist;                      this is an optional line, data must be SORTed first
PLOT request-list < / options > ;
```

Options for the PROC PLOT statement include:

UNIFORM	scales axes of all plots produced by this PLOT procedure uniformly
HPERCENT= <i>value</i>	specifies the percentage of the horizontal page to use for each plot. <i>Values</i> less than 100 will yield reduced plot. <i>Values</i> greater than 100 will spread a plot across more than one page. Can also use HPCT= <i>value</i> .
VPERCENT= <i>value</i>	specifies the percentage of the vertical page to use for each plot. <i>Values</i> follow the same rule as those for HPERCENT. Can also use VPCT= <i>value</i> .

More than one PLOT statement can be used. This is beneficial if the options for each PLOT statement are going to be different. Options for the PLOT statement include:

HAXIS= <i>values</i>	specifies the values for tick marks on the horizontal axis
VAXIS= <i>values</i>	specifies the values for tick marks on the vertical axis
HREF= <i>values</i>	draws lines on the plot perpendicular to the horizontal axis at the points specified
VREF= <i>values</i>	draws lines on the plot perpendicular to the vertical axis at the points specified
BOX	draws a border around the entire plot
OVERLAY	overlays all plots specified in the plot statement on one set of axes

The default plotting symbol is the letter A for a single point. For points that coincide, a letter B is used for two letters that coincide on the plot, C for three, and so on through the alphabet. A specific plotting symbol can be used or the value of another variable. Each of these will be demonstrated in the following objectives.

In a PLOT statement, the request list of plots can take various forms. Here are some examples.

PLOT y*x;	generates a plot in which Y plotted on the vertical axis and X is plotted on the horizontal axis
PLOT a*b a*c; PLOT a*(b c);	generates two plots in which A is plotted on the vertical axes of both plots.
PLOT (a b)*(c d);	generates four plots. A by C, A by D, B by C, B by D
PLOT a*b='x';	generates a plot of A by B where x is the plotting symbol. (This illustrates how one can select alternative plotting symbols.)

Objective 1: Enter the follow program in the SAS editor. Produce two full-page plots: one of w (vertical) by x (horizontal) and the second of y by z.

```
DM 'LOG; CLEAR; ODSRESULTS; CLEAR; ' ;

DATA one;
INPUT w x y z ;
DATALINES;
1 3 12 19
1 3 15 21
1 5 18 25
2 9 21 14
2 11 27 17
2 19 32 24
;
PROC PLOT DATA=one;
PLOT w*x y*z ;
RUN;
QUIT;
```

Objective 2: Modify the plots in Objective 1 so that the points in the w by x plot are plotted using the # symbol, and the points in the y by z plot are the values of the variable w.

```
PROC PLOT DATA=one;
PLOT w*x='#' y*z=w;
```

Objective 3: Repeat Objective 1 except place both of the plots on the same page. Add VPERCENT=50 option to the PLOT procedure. Try HPERCENT=50. Try both at once.

Objective 4: Keep VPERCENT=50 in the first Objective. Modify the procedure so that two PLOT statements are created; each statement with different options.

```
PLOT w*x / VREF = 1 2 ; PLOT y * z / BOX ;
```

Objective 5: Plot x by y and x by z on the same set of axes.

```
PLOT x * (y z) / OVERLAY ;
```

Objective 6: Note that in Objective 5 the Y and Z values are not marked distinctly from each other. Modify the previous plot statement so that the plot distinguishes the Y responses from the Z responses.

```
PLOT x * y = 'Y' x * z = 'Z' / OVERLAY ;
```

Clearly the PLOT procedure does not produce high-resolution graphics. The advantage to using the PLOT procedure is that it is a quick analytical tool. For final presentations, a better quality image is usually preferable. The SAS/GRAPH equivalent to the PLOT procedure is the GPLOT procedure. In GPLOT there are many more options than are available in the PLOT procedure. The GPLOT procedure will be covered in a later section. SGPLOT is a newer procedure which will be covered in a later section also.