DATA Step Information – Indicator Function

Within the DATA step new variables can be created or existing variables modified using variable assignment statements. Additionally, IF-THEN or IF-THEN-ELSE statements can be used within the DATA step to perform these sorts of operations also. Here, an indicator function is presented.

An indicator function, in general, is a function that takes a value of 1 when a specified condition is met, and a value of 0 otherwise. The indicator function is a condition enclosed in parentheses. This is illustrated in the following simple SAS program.

```
DATA  one;
INPUT group $  w x y z ;
example1 = (group = "A") ;
example2 = (INT(w/2) = w/2 ) ;
example3 = (x >= 30) ;
example4 = ( (group ne "A") OR (y < 100) ) ;
example5 = ( (group ne "A") AND (y < 100) ) ;
example6 = z + 5*(group = "A") + 10*(group = "B") ;
DATALINES ;
A  17  35  104  79
A  19  32   90  92
B  16  39  101  89
B  21  40   95  85
C  12  29   88  81
C  16  27   84  83
;
PROC PRINT DATA=one NOOBS ;
RUN;
QUIT;
```

group	w	x	y	z	example1	example2	example3	example4	example5	example6
A	17	35	104	79	1	0	1	0	0	84
A	19	32	90	92	1	0	1	1	0	97
B	16	39	101	89	0	1	1	1	0	99
B	21	40	95	85	0	0	1	1	1	95
C	12	29	88	81	0	1	0	1	1	81
C	16	27	84	83	0	1	0	1	1	83

In example1 a value of 1 is assigned to those observations in group A, and 0 is assigned to those outside of group A. To do this using IF-THEN-ELSE statements, one could have written

```
IF group = "A" then example1 = 1; ELSE example1 = 0 ;
```

Without the ELSE statement, when the group was different from A, the value of example1 would have been missing.

Example 2 identifies with a 1 those values of w that are even numbers. If the greatest integer, INT( ) function, of the quantity w divided by 2 is the same as w divided by 2, then the value of w must be even. Note the value of w divided by 2 is not assigned to a variable in the data table one. Like Example1, this variable assignment could have been completed using IF-THEN-ELSE statements.

```
IF INT(w/2) = w/2 THEN example2 = 1; ELSE example2 = 0;
```

Example3 is a dichotomous variable that identifies when the value of x is at least 30. Equivalently, IF-THEN-ELSE statements could have been used:

```
IF w >= 30 THEN example3 = 1; ELSE example3 = 0 ;
```

Example4 and Example5 illustrate that more than one condition can be specified within an indicator function. Note the differences in the values of the two variables due to the differences between the logical OR and the logical AND in the indicator function. OR implies that either or both of the conditions must be true to assign a 1; AND implies that both conditions must be true to assign a 1.

Example6 is the value of the variable z only for group C; for group A example6 is z + 5; and for group B example6 is z + 10.

It's been demonstrated here that IF-THEN-ELSE statements can achieve the same results as an indicator function. Note that the syntax for the indicator functions for these examples tends to be simpler.