The FORMAT Procedure

The FORMAT procedure allows you to translate values entered as data to more meaningful values or terms. For example, a variable in a data set may have values 1 and 2. These values represent the responses YES and NO, respectively, to a survey question. (This is different from the LABEL statement used in a DATA step. Recall, LABEL allows to more fully define the variable name, not the actual values the variable can take.)

When you use the format procedure, the formats defined are active during the entire SAS session; that is, until you exit SAS. Whenever you run a procedure and wish to have the formatted values used in the output, you must use a FORMAT statement in the procedure.

You can format either numeric or character values. PROC FORMAT produces no output. You call the formats you define into procedures after you have defined the formats.

The syntax of the FORMAT procedure:

```
PROC FORMAT;
PICTURE      name   range-1 = 'picture-1'
                    .
                    .
                    .
                    range-n = 'picture-n';

VALUE        name   range-1 = 'formatted-value-1'
                    .
                    .
                    .
                    range-n = 'formatted-value-n';

VALUE       $name   'range-1' = 'formatted-value-1'
                    .
                    .
                    .
                    'range-n' = 'formatted-value-n';
```

The PICTURE statement defines a template for printing numbers. For example, a special format for phone numbers or zip code + 4.

The VALUE statement defines a format that writes a variables' value as a different value. The first value statement appropriate for numeric values of range. When the values of range are character values, the form of the second variable statement ($name) should be used.

When a formatted value is called into a procedure, it is done in a format statement. Format variable names are followed by a period when they are called in a FORMAT statement. The period is not used when the format variable is created. The syntax of the FORMAT statement is:

FORMAT variable1 format1. - - variablen formatn. ;

Not all variables "used" by a procedure need to be formatted. Some formats can apply to more than one variable. If a format applies to several variables, each must be listed in the FORMAT statement followed by the format variable.
The FORMAT procedure has much more flexibility than what is shown in the demonstration below. For all of the information about PROC FORMAT, see *SAS Procedures Guide.*

Run the following program. Carefully examine the output for what each of the formats does.

```
data one;
input id exp salary region gender bs ms phd group $;
salary2 = salary*1000;
cards;
000171831 8.5  41.5  1  1 1 1 2 E
777889999 10   53.4  3  2 1 1 1 E
111223333 13   65.0  4  . 1 1 2 B
222334444 20   75.0  2  3 2 1 1 S
;
proc format;

picture a other = '000-00-0000' ;
picture b other = '000.00'  ;
picture c other = '000000' (prefix='$') ;

value salfmt low-49.9 = 'Below 50,000'
             50.0-59.9 = '50,000 - 60,000'
             60-high  = '60,000 and over';

value regfmt 1 = 'Northwest'
             2 = 'Central'
            3-4= 'Southern';

value gen 1 = 'Male'
          2 = 'Female'
          . = 'Missing value'
      other = 'Miscoded';

value degree 1 = 'Yes'
             2 = 'No';

value degrus 1 = 'Da'
             2 = 'Nyet';

value $grp     "E" = "Engineering"
               "B" = "Business"
               "S" = "Science" ;

proc print data=one;
format id a. salary b. ;

proc print data=one;
format id a. salary salfmt. salary2 c. region regfmt. gender gen.
       bs degree. ms degree.  phd degrus.  group $grp. ;
```

```
proc print data=one;
format bs ms phd degree. ;

run;
quit;
```